Chapter 8

CYLINDRICAL ANALYSIS PROBLEMS

8.1 Introduction

We will slightly increase the prior one-dimensional examples by extending them to problems formulated in cylindrical coordinates. There are many problems that can accurately be modeled as being revolved about an axis. Many of these can be analyzed by employing a radial coordinate, r, and an axial coordinate, z. Solids of revolution can be formulated in terms of the two-dimensional area that is revolved about the axis. Numerous objects of that type are also very long in the axial direction and can be treated as segments of an infinite cylinder. This reduces the analysis to a one-dimensional study in the radial direction. We will begin with that common special case. We will find that changing to these *cylindrical coordinates* will make small changes in the governing differential equations and the corresponding integral theorems that govern the finite element formulation. Also the volume and surface integrals take on special forms. These use the *Theorems of Pappus*. The first states that the surface area of a revolved arc is the product of the arc length and the distance traveled by the centroid of the arc. The second states that the volume of revolution of the generating area is the product of the area and the distance traveled by its centroid. In both cases the distance traveled by the centroid, in full revolution, is $2\pi\bar{r}$ where $\bar{r}$ is the centroid radial coordinate of the arc or area. If we consider differential arcs or areas, then the corresponding differential surface or volume of revolution are $dA = 2\pi r\, dL$ and $dV = 2\pi r\, dr\, dz$.

8.2 Heat Conduction in a Cylinder

The previous one-dimensional heat transfer model becomes slightly more complicated here. When we consider a point on a radial line we must remember that it is a cross-section of a ring of material around the hoop of the cylinder. Thus as heat is conducted outward in the radial direction it passes through an ever increasing amount of material. The resulting differential equation for thermal equilibrium is well known:

$$\frac{1}{r}\frac{d}{dr}\left(r\,k\,\frac{d\theta}{dr}\right)+Q\;=\;0 \tag{8.1}$$

where r is the radial distance from the axis of revolution, k is the thermal conductivity, θ is the temperature, and Q is the internal heat generation per unit volume. One can have essential boundary conditions where θ is given or as a surface flux condition

$$-\,r\,k\,\frac{d\theta}{dr}\;=\;q \tag{8.2}$$

where q is the flux normal to the surface (i.e., radially). If we multiply Eq. 8.1 by r, it would look like our previous one-dimensional form:

$$\frac{d}{dr}\left(k*\frac{d\theta}{dr}\right)+Q*\;=\;0$$

where $k* = r\,k$ and $Q* = r\,Q$ could be viewed as variable coefficients. This lets us find the required integral (variational) form by inspection. It is

$$I\;=\;2\pi\Delta z\int_L \tfrac{1}{2}\Big[\;k*(d\theta/dr)^2-Q*T\;\Big]\,r\,dr\qquad\rightarrow\min \tag{8.3}$$

where the integration limits are the inner and outer radii of the cylindrical segment under study. The typical length in the axial direction, Δz, is usually defaulted to unity. The corresponding element square conduction matrix is

$$\mathbf{S}^e\;=\;2\pi\int_{L^e} k^e\;\frac{d\mathbf{H}^{e^T}}{dr}\;\frac{d\mathbf{H}^e}{dr}\;r\,dr \tag{8.4}$$

and the source vector (if any) is

$$\mathbf{C}^e_Q\;=\;2\pi\int_{L^e}\mathbf{H}^{e^T}\;Q^e\;r\,dr\,. \tag{8.5}$$

If we consider a two node (linear) line element in the radial direction we can use our previous results to write these matrices by inspection. Noting that $L^e = (r_2 - r_1)^e$ and assuming a constant material property, k, in the element gives

$$\mathbf{S}^e\;=\;2\pi\,\frac{k^e}{(L^e)^2}\begin{bmatrix}1 & -1\\ -1 & 1\end{bmatrix}\int_{r_1}^{r_2} r\,dr$$

$$\mathbf{S}^e\;=\;2\pi\,\frac{k^e(r_2^2-r_1^2)^e}{2(L^e)^2}\begin{bmatrix}1 & -1\\ -1 & 1\end{bmatrix}=\pi\,\frac{k^e(r_2+r_1)^e}{(r_2-r_1)^e}\begin{bmatrix}1 & -1\\ -1 & 1\end{bmatrix}. \tag{8.6}$$

Thus, unlike the original one-dimensional case the conduction matrix depends on where the element is located, that is, it depends on how much material it includes (per unit length in the axial direction).


```
! ..................................................... ! 1
!  **  ELEM_SQ_MATRIX PROBLEM DEPENDENT STATEMENTS FOLLOW ** ! 2
! ..................................................... ! 3
!  CYLINDRICAL HEAT CONDUCTION, (see Section 8.2)       ! 4
                                                        ! 5
 REAL(DP), PARAMETER :: TWO_PI = 6.2831853072d0         ! 6
 REAL(DP)            :: CONST, DET                      ! 7
 REAL(DP)            :: K_RR, SOURCE                    ! 8
 INTEGER             :: IP                              ! 9
                                                       !10
! 1/R * d[R K_RR dT/dR]/dR + Q = 0, Example 109         !11
                                                       !12
! PROP_1 = CONDUCTIVITY K_RR                           !13
! PROP_2 = SOURCE PER UNIT VOLUME, Q                   !14
                                                       !15
!-->  DEFINE ELEMENT PROPERTIES                        !16
  K_RR = GET_REAL_LP (1) ; SOURCE = GET_REAL_LP (2)    !17
  E (1, 1) = K_RR ! CONSTITUTIVE                       !18
                                                       !19
!      STORE NUMBER OF POINTS FOR FLUX CALCULATIONS    !20
  CALL STORE_FLUX_POINT_COUNT ! Save LT_QP             !21
                                                       !22
!-->   NUMERICAL INTEGRATION LOOP                      !23
   DO IP = 1, LT_QP                                    !24
     H   = GET_H_AT_QP (IP)   ! INTERPOLATION FUNCTIONS !25
     XYZ = MATMUL (H, COORD)  ! FIND RADIUS (R)        !26
     DLH = GET_DLH_AT_QP (IP) ! FIND LOCAL DERIVATIVES !27
     AJ = MATMUL (DLH, COORD) ! FIND JACOBIAN AT THE PT !28
!         FORM INVERSE AND DETERMINATE OF JACOBIAN     !29
     CALL INVERT_JACOBIAN (AJ, AJ_INV, DET, N_SPACE)   !30
     CONST  = TWO_PI * DET * WT(IP) * XYZ (1) ! 2 PI*|J|*w*R !31
                                                       !32
!         EVALUATE GLOBAL DERIVATIVES, DGH == B        !33
     DGH = MATMUL (AJ_INV, DLH)                        !34
     B   = COPY_DGH_INTO_B_MATRIX (DGH) ! B = DGH      !35
                                                       !36
     C = C + CONST * SOURCE * H ! VOLUMETRIC SOURCE OPTION !37
                                                       !38
!      CONDUCTION SQUARE MATRIX                        !39
     S = S + CONST * MATMUL ((MATMUL (TRANSPOSE (B), E)), B) !40
                                                       !41
!-->  SAVE COORDS, E AND B MATRIX, FOR POST PROCESSING !42
     CALL STORE_FLUX_POINT_DATA (XYZ, E, B)            !43
   END DO                                              !44
! ** END ELEM_SQ_MATRIX PROBLEM DEPENDENT STATEMENTS ** !45
```

Figure 8.2.1 Numerically integrated cylindrical heat transfer

Next, we assume a constant source term so the source vector becomes

$$\mathbf{C}_Q^e \;=\; 2\pi\, Q^e \int_{L^e} \mathbf{H}^{e^T}\, r\, dr\,.$$

But the varying radial position, r, must also be accounted for in the integration. One approach to this integration is to again use our isoparametric interpolation and let $r = \mathbf{H}^e\, \mathbf{R}^e$, where $\mathbf{R}^e$ notes the radial position of each node (i.e., input data). Then

$$\mathbf{C}_Q^e \;=\; 2\pi\, Q^e \int_{L^e} \mathbf{H}^{e^T}\, \mathbf{H}^e \; dr \; \mathbf{R}^e \,.$$

For a general element type we would have to evaluate Eq. 8.7 by numerical integration. Then the summation is

$$\mathbf{C}_Q^e \;=\; 2\pi \sum_q^{n_q} \mathbf{H}_q^{e^T}\, \mathbf{H}_q^e\, r_q\, Q_q\, |\mathbf{J}_q^{\,e}| w_q$$

where we have interpolated for the position, $r_q = \mathbf{H}_q^e \mathbf{R}^e$ and allowed for a spatially varying source input at the nodes with $Q_q = \mathbf{H}_q^e \mathbf{Q}^e$. A similar express would evaluate the the square matrix in Eq. 8.4. Returning to the linear line element, we have previously exactly evaluated the matrix product integral (mass matrix) and can write the resultant constant source term as

$$\mathbf{C}_Q^e \;=\; \frac{2\pi Q^e L^e}{6} \begin{bmatrix} 2 & 1 \\ 1 & 2 \end{bmatrix} \begin{Bmatrix} r_1 \\ r_2 \end{Bmatrix}^e \;=\; \frac{2\pi Q^e L^e}{6} \begin{bmatrix} 2r_1 + r_2 \\ r_1 + 2r_2 \end{bmatrix}^e . \tag{8.7}$$

Note that the result depends on the element location because the source is being created in a larger volume of material as the radius increases.

As a simple numerical example consider a cylinder with constant properties, no internal heat generation, an inner radius temperature of $\theta = 100$ at $r = 1$, and an outer radius temperature of $\theta = 10$ at $r = 2$. Select a model with four equal length elements and five nodes. Numbering the nodes radially we have essential boundary conditions of $\theta_1 = 100$ and $\theta_5 = 10$. Considering the form in Eq. 8.6, we note that the element values of $(r_2 + r_1)^e / L^e$ are 9, 11, 13, and 15, respectively. Therefore, we can write the assembled system equations as

$$\pi k^e \begin{bmatrix} 9 & -9 & 0 & 0 & 0 \\ -9 & (9+11) & -11 & 0 & 0 \\ 0 & -11 & (11+13) & -13 & 0 \\ 0 & 0 & -13 & (13+15) & -15 \\ 0 & 0 & 0 & -15 & 15 \end{bmatrix} \begin{Bmatrix} \theta_1 \\ \theta_2 \\ \theta_3 \\ \theta_4 \\ \theta_5 \end{Bmatrix} = \begin{Bmatrix} q_1 \\ 0 \\ 0 \\ 0 \\ -q_5 \end{Bmatrix}.$$

Applying the essential boundary conditions, and dividing both sides by the leading constant gives the reduced set

$$\begin{bmatrix} 20 & -11 & 0 \\ -11 & 24 & -13 \\ 0 & -13 & 28 \end{bmatrix} \begin{Bmatrix} \theta_2 \\ \theta_3 \\ \theta_4 \end{Bmatrix} = 100 \begin{Bmatrix} 9 \\ 0 \\ 0 \end{Bmatrix} + 10 \begin{Bmatrix} 0 \\ 0 \\ 15 \end{Bmatrix}.$$

Solving yields the internal temperature distribution of $\theta_2 = 71.06$, $\theta_3 = 47.39$, and $\theta_4 = 27.34$. Comparing with the exact solution of $\theta = [\theta_1 \ln(r_5/r) + \theta_5 \ln(r/r_1)] / \ln(r_5/r_1)$ shows that our approximation is accurate to at least three significant figures. Also note that both the exact and approximate temperature distributions are independent of the thermal conductivity k. This is true only because the internal heat generation Q was zero. Of course, k does have some effect on the two external heat fluxes (thermal


```
title 'Cylindrical Heat Transfer, Sec. 8.2'           ! 1
axisymmetric  ! Problem is axisymmetric, x radius     ! 2
example   109 ! Source library example number         ! 3
data_set   01 ! Data set for example (this file)      ! 4
nodes       5 ! Number of nodes in the mesh           ! 5
elems       2 ! Number of elements in the system      ! 6
dof         1 ! Number of unknowns per node           ! 7
el_nodes    3 ! Maximum number of nodes per element   ! 8
space       1 ! Solution space dimension              ! 9
b_rows      1 ! Number of rows in the B matrix        !10
shape       1 ! Element shape, 1=line, 2=tri, 3=quad  !11
remarks     7 ! Number of user remarks                !12
gauss       3 ! Maximum number of quadrature point    !13
el_real     2 ! Number of real properties per element !14
el_homo       ! Element properties are homogeneous    !15
no_error_est  ! No SCP element error estimates        !16
quit ! keyword input, remarks follow                  !17
1 1/r * d[r K_rr dT/dr]/dr + Q = 0, Example 109       !18
2 Real FE problem properties are:                     !19
3 K_rr = GET_REAL_LP (1) conductivity                 !20
4 Q    = GET_REAL_LP (2) source per unit length       !21
5 Mesh T=100, r=1 *----*----*----*----* r= 2, T=10    !22
6 Nodes, (Elem)   1    2(1) 3    4(2) 5, K = 1, Q = 0 !23
7 T = [T_1*ln(r_5/r) + T_5*ln(r/r_1)]/ln(r_5/r_1)     !24
1    1   1.     ! begin nodes flag x                  !25
2    0   1.25                                        !26
3    0   1.50   ! note exact T_3=47.35               !27
4    0   1.75                                        !28
5    1   2.0                                         !29
 1    1    2    3   ! begin elements                 !30
 2    3    4    5                                   !31
1 1 100.            ! essential bc                    !32
5 1 10.                                               !33
 1 1. 0.            ! Elem K Q                        !34
```

Figure 8.2.2 Typical data for cylindrical conduction

```
T = [T_1*ln(r_5/r) + T_5*ln(r/r_1)]/ln(r_5/r_1) ! 1
                                                ! 2
*** REACTION RECOVERY ***                       ! 3
   NODE, PARAMETER,     REACTION, EQUATION      ! 4
      1,  DOF_1,       8.1591E+02     1         ! 5
      5,  DOF_1,      -8.1591E+02     5         ! 6
                                                ! 7
***  OUTPUT OF RESULTS IN NODAL ORDER  ***      ! 8
    NODE,  Radius r,   DOF_1,                   ! 9
       1  1.0000E+00  1.0000E+02                !10
       2  1.2500E+00  7.1046E+01                !11
       3  1.5000E+00  4.7356E+01                !12
       4  1.7500E+00  2.7344E+01                !13
       5  2.0000E+00  1.0000E+01                !14
```

Figure 8.2.3 Results from two quadratic axisymmetric elements



Figure 8.2.4 Temperatures for eight linear elements



Figure 8.2.5 Exact temperature error



Figure 8.2.6 Flux per unit area for eight linear elements

reactions), q_1 and q_5, necessary to maintain the two prescribed surface temperatures. Substituting back into the first equation to recover the thermal reaction we obtain

$$\pi\,[\ 9(100) - 9(71.06) + 0\ ]\ =\ 818.3\ =\ q_1 / k^e$$

entering at the inner radius. This compares quite well with the exact value of 8.15.8. The fifth equation gives q_5 an equal amount exiting at the outer radius. Therefore, in this problem the heat flux is always in the positive radial direction.

It should be noted that if we had used a higher order element then the integrals would have been much more complicated than the one-dimensional case. This is typical of most axisymmetric problems. Of course, in practice we use numerical integration to automate the evaluation of the element matrices, as described above. A typical implementation is shown in Fig. 8.2.1. A new consideration is that during the intergration we must include the radius. This is done in line 31 using the radial position interpolated in line 26. The data for using two quadratic (3 noded) line elements to solve the above example is shown in Fig. 8.2.2, and the results are summarized in Fig. 8.2.3. There we see that the center temperature, line 12, is accurate to four significant figures and that the temperatures and reaction fluxes compare closely to the above four linear element model. The keyword "axisymmetric" (line 2 of Fig. 8.2.2) is not really used in formulating the element matrices as we have hard coded that knowledge in Fig. 8.2.1 but



Figure 8.2.7 Estimated energy norm error



Figure 8.2.8 Exact energy norm error


```
! T = [T_in*ln(r_out/r) + T_out*ln(r/r_in)]/ln(r_out/r_in)  ! 1
                                                           ! 2
*** REACTION RECOVERY ***                                  ! 3
   NODE, PARAMETER,     REACTION, EQUATION                 ! 4
      1,  DOF_1,       8.1640E+02     1                    ! 5
      9,  DOF_1,      -8.1640E+02     9                    ! 6
                                                           ! 7
***  OUTPUT OF RESULTS AND EXACT VALUES IN NODAL ORDER  *** ! 8
    NODE,  Radius r,   DOF_1,      EXACT1,                 ! 9
       1  1.0000E+00  1.0000E+02  1.0000E+02               !10
       2  1.1250E+00  8.4714E+01  8.4707E+01               !11
       3  1.2500E+00  7.1036E+01  7.1026E+01               !12
       4  1.3750E+00  5.8662E+01  5.8651E+01               !13
       5  1.5000E+00  4.7363E+01  4.7353E+01               !14
       6  1.6250E+00  3.6968E+01  3.6960E+01               !15
       7  1.7500E+00  2.7344E+01  2.7338E+01               !16
       8  1.8750E+00  1.8383E+01  1.8380E+01               !17
       9  2.0000E+00  1.0000E+01  1.0000E+01               !18
                                                           !19
** SUPER_CONVERGENT AVERAGED NODAL FLUXES & EXACT FLUXES ** !20
NODE,   Radius r,   FLUX_1,     EXACT1   (per unit area)   !21
   1  1.000E+00  1.252E+02   1.298E+02                     !22
   2  1.125E+00  1.144E+02   1.154E+02                     !23
   3  1.250E+00  1.042E+02   1.039E+02                     !24
   4  1.375E+00  9.426E+01   9.443E+01                     !25
   5  1.500E+00  8.645E+01   8.656E+01                     !26
   6  1.625E+00  7.982E+01   7.990E+01                     !27
   7  1.750E+00  7.408E+01   7.420E+01                     !28
   8  1.875E+00  6.964E+01   6.925E+01                     !29
   9  2.000E+00  6.575E+01   6.492E+01                     !30
                                                           !31
----------------------------------------                   !32
             ERROR IN        % ERROR IN                    !33
ELEMENT,     ENERGY_NORM,    ENERGY_NORM,                  !34
----------------------------------------                   !35
      1      3.7078E+00      1.3710E+00                    !36
      2      2.8507E+00      1.0541E+00                    !37
      3      2.9117E+00      1.0766E+00                    !38
      4      2.3987E+00      8.8695E-01                    !39
      5      2.1187E+00      7.8341E-01                    !40
      6      1.9101E+00      7.0629E-01                    !41
      7      1.5382E+00      5.6878E-01                    !42
      8      1.5794E+00      5.8401E-01                    !43
```

Figure 8.2.9 Exact and computed results in conducting cylinder



Figure 8.3.1 Hoop strain due to radial displacement


```
! B       = STRAIN-DISPLACEMENT MATRIX, (N_R_B, LT_FREE)       ! 1
! BODY    = BODY FORCE VECTOR, (N_SPACE)                       ! 2
! DGH     = GLOBAL DERIVS OF FUNCTIONS H, (N_SPACE, LT_N)      ! 3
! LT_FREE = NUMBER OF DEGREES OF FREEDOM PER ELEMENT           ! 4
! LT_N    = MAXIMUM NUMBER OF NODES FOR ELEMENT TYPE           ! 5
! N_R_B   = NUMBER OF ROWS IN B AND E MATRICES                 ! 6
! .....................................................       ! 7
!  **  ELEM_SQ_MATRIX PROBLEM DEPENDENT STATEMENTS FOLLOW **   ! 8
! .....................................................       ! 9
!  Cylindrical Stress Analysis, (see Section 8.3)              !10
                                                              !11
 REAL(DP), PARAMETER :: TWO_PI = 6.2831853072d0               !12
 REAL(DP)            :: CONST, DET                            !13
 REAL(DP)            :: E_mod, P_ratio, Rho, Spin             !14
 INTEGER             :: IP                                    !15
                                                              !16
! Elem real prop: Young's modulus, Poisson's ratio, Density   !17
! Misc real prop: Spin about z-axis, radians per second       !18
  E_mod = GET_REAL_LP (1) ; P_ratio = GET_REAL_LP (2)         !19
  Rho   = GET_REAL_LP (3) ; Spin    = GET_REAL_MISC (1)       !20
                                                              !21
!               CONSTITUTIVE LAW                            !22
  E(1,1) = E_mod*(1 - P_ratio)/((1 + P_ratio)*(1 - 2*P_ratio)) !23
  E(2,1) = E_mod*P_ratio/((1 + P_ratio)*(1 - 2*P_ratio))      !24
  E(1,2) = E(2,1)  ;  E(2,2) = E(1,1)                         !25
                                                              !26
!      STORE NUMBER OF POINTS FOR FLUX CALCULATIONS           !27
  CALL STORE_FLUX_POINT_COUNT ! Save LT_QP                    !28
                                                              !29
!-->   NUMERICAL INTEGRATION LOOP                            !30
   DO IP = 1, LT_QP                                           !31
     H   = GET_H_AT_QP (IP)   ! INTERPOLATION FUNCTIONS       !32
     XYZ = MATMUL (H, COORD)  ! FIND RADIUS (ISOPARAMETRIC)   !33
     DLH = GET_DLH_AT_QP (IP) ! FIND LOCAL DERIVATIVES        !34
     AJ = MATMUL (DLH, COORD) ! FIND JACOBIAN AT THE PT       !35
!         FORM INVERSE AND DETERMINATE OF JACOBIAN           !36
     CALL INVERT_JACOBIAN (AJ, AJ_INV, DET, N_SPACE)          !37
     CONST = TWO_PI * DET * WT(IP) * XYZ (1) ! 2 PI*|J|*w*R   !38
                                                              !39
!      EVALUATE GLOBAL DERIVATIVES & STRAIN-DISPLACEMENT      !40
     DGH = MATMUL (AJ_INV, DLH)                               !41
     B (1, :) = DGH (1, :)       ! DU/DR radial strain        !42
     B (2, :) = H (:) / XYZ (1)  ! U/R hoop strain            !43
                                                              !44
!      STIFFNESS MATRIX                                       !45
     S = S + CONST * MATMUL ((MATMUL (TRANSPOSE (B), E)), B)  !46
                                                              !47
     BODY = - Rho * XYZ (1) * Spin **2 ! -Rho R Omega^2       !48
     C    = C + CONST * BODY (1) * H   ! CENTRIFUGAL RESULT   !49
                                                              !50
!-->  SAVE COORDS, E AND B MATRIX, FOR POST PROCESSING        !51
     CALL STORE_FLUX_POINT_DATA (XYZ, E, B)                   !52
   END DO                                                     !53
! ** END ELEM_SQ_MATRIX PROBLEM DEPENDENT STATEMENTS **       !54
```

Figure 8.3.2 Cylindrical stress with centrifugal loads

it is used in the geometric properties (volume and centroid) calculations that are provided for data validation and it would have triggered necessary calculations in the error estimation stage if keyword "no_error_est" had not been present. That overrides the default state of including error estimates, which is done in the next example.

As another comparison, now including error estimates, consider the linear element model run with eight equal elements. The nodal results are still accurate to about four significant figures but, of course, are less accurate than that inside the elements where the logarithmic exact value is being approximated by a straight line. The actual temperature values are shown in Fig.8.2.4 along with the exact temperature error in Fig. 8.2.5. Because of the boundary conditions used here remember that the temperatures are independent of the conductivity, but the flux is not. For a planar wall of the same thickness and same inner and outer wall essential boundary conditions the temperature would have been exactly linear with position and the total heat flux and the heat flux per unit area would have been constant. Here we see in Fig. 8.2.6 that the smoothed heat flux per unit area is not constant but drops off radially because it is passing through more material (a larger area) as the radius increases. The exact flux per unit area is given by $k(\theta_{in} - \theta_{out}) \,/\, [r\, ln(r_{out} \,/\, r_{in})]$.

The smoothed fluxes, per unit area, are compared to the element piecewise constant values in order to develop an error estimate. The estimated error, in the energy norm, is shown in Fig. 8.2.7. It was obtained by averaging the element values at the nodes. In this simple problem we have the exact solution so we can compare the estimate with the corresponding exact values in Fig. 8.2.8. Here we see that the SCP energy norm error estimate agrees with the exact energy norm error reasonably well. Referring back to Fig. 8.2.5 we observe that spatial distribution of the exact function error and the exact energy norm error can be quite different. The error in the solution is always exactly zero at nodes with essential boundary conditions. However, small distances away from such locations the function value error can increase very rapidly. In Fig. 8.2.9 we see (lines 5 and 6) that the heat flux reaction, over a total $2\pi r$ section, has changed very little. Remember that such reaction fluxes are obtained from the integral form and thus are more accurate than element fluxes which are extrapolated to the boundary. This can be seen again in the averaged nodal fluxes, per unit area, (lines 22-30) which when multiplied by the section areas give inner and outer values (which should be the same) of 787 and 826 for an average of 807 which represents an error of about one percent.

8.3 Cylindrical Stress Analysis

Another common problem is the analysis of an axisymmetric solid with axisymmetric loads and supports. This becomes a two-dimensional analysis that is very similar to plane strain analysis. The radial and axial displacement components will be denoted by u and v. These are the same unknowns used in the plane strain study. In addition to the previous strains there is another strain known as the *hoop strain*, ε_t, and a corresponding hoop stress, σ_t. The hoop strain results from the change in length of a fiber of material around the circumference of the solid. The definition of a normal strain is a change in length divided by the original length. The circumference at a typical radial position is $L = 2\pi r$. When such a point undergoes a radial displacement of u it occupies a new radial position of $(r + u)$, as shown in Fig. 8.3.1. It has a corresponding increase in circumference. The hoop strain becomes

$$\varepsilon_t \;=\; \frac{\Delta L}{L} = \frac{2\pi(r+u) - 2\pi r}{2\pi r} = \frac{u}{r}\,. \tag{8.8}$$

Then, our strains to be computed for the cylindrical analysis are denoted as

$$\varepsilon^T = [\varepsilon_r \quad \varepsilon_t] \tag{8.9}$$

and the corresponding stress components are

$$\sigma^T = [\sigma_r \quad \sigma_t]. \tag{8.10}$$

For an isotropic material the stress-strain law is like that for the plane strain case. In the axial, or z, direction of the assumed infinite cylinder one can either set the stress or strain to zero, but not both. They are related by:

$$\varepsilon_z = \frac{1}{E}(\sigma_z - \nu\sigma_r - \nu\sigma_t). \tag{8.11}$$

There is no shear stress or strain in the cylindrical case. In the case of an infinite cylinder the above formulation simplifies since $\nu = 0$ and $\partial/\partial z = 0$. Thus, we consider only the radial displacement, u, and the strains and stresses in the radial and hoop directions. This gives the two strain-displacement relations as

$$\boldsymbol{\varepsilon} = \mathbf{B}^e \mathbf{u}^e \tag{8.12}$$

where

$$\mathbf{B}^e = \begin{bmatrix} \partial\mathbf{H}/\partial r \\ \mathbf{H}/r \end{bmatrix}. \tag{8.13}$$

Note that this is the first time we have computed two secondary items (strains) from a single primary unknown. Now the stresses and the stress-strain law must have the same number of rows. The radial and hoop stresses are $\boldsymbol{\sigma} = \mathbf{D}\,\boldsymbol{\varepsilon}$ where

$$\mathbf{D} = \begin{bmatrix} D_{11} & D_{12} \\ D_{21} & D_{22} \end{bmatrix} \tag{8.14}$$

and for an isotropic material $D_{11} = D_{22} = E(1-\nu)/(1+\nu)(1-2\nu)$, while the off-diagonal terms are $D_{12} = D_{21} = E\nu/(1+\nu)(1-2\nu)$. Note that this constitutive law encounters problems for an "incompressible" material where $\nu = 1/2$ and division by zero occurs. That forces one to employ an alternate theory. The stiffness matrix,

$$\mathbf{K}^e = 2\pi \int_{A^e} \mathbf{B}^{e^T} \mathbf{D}^e \mathbf{B}^e \, r \, dr \, dz \,,$$

can be expanded to the form

$$\begin{aligned} \mathbf{K}^e = 2\pi\,\Delta z \int_{L^e} \Bigg[& D_{11} \frac{\partial \mathbf{H}^T}{\partial r} \frac{\partial \mathbf{H}}{\partial r} \\ & + D_{12}\left(\frac{\partial \mathbf{H}^T}{\partial r}\mathbf{H} + \mathbf{H}^T \frac{\partial \mathbf{H}}{\partial r} \right)/r + D_{22}\mathbf{H}^T\mathbf{H}/r^2 \Bigg] r\, dr\,. \end{aligned} \tag{8.15}$$

The first integral we just evaluated and is given in Eqs. 8.4 and 8.6 if we let $\Delta z = 1$ and replace k with D_{11}. The second term we integrate by inspection since the r terms cancel. The result is

$$\mathbf{K}_{12}^e = 2\pi\ \Delta z\ D_{12}^e \begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix}. \tag{8.16}$$

The remaining contribution is more difficult since it involves division by r. Assuming constant D_{22} we have

$$\mathbf{K}_{22} = 2\pi\ \Delta z\ D_{22} \int_{L^e} \frac{1}{r}\, \mathbf{H}^T \mathbf{H}\, dr \tag{8.17}$$

which requires analytic integration involving logarithms, or numerical integration. Using a one point (centroid) quadrature rule gives the approximation

$$\mathbf{K}_{22}^e = \frac{2\pi\ \Delta z\ D_{22}^e L^e}{2(r_1 + r_2)^e} \begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}. \tag{8.18}$$

For a cylinder the loading would usually be a pressure acting on an outer surface or an internal centrifugal load due to a rotation about the z-axis. For a pressure load the resultant force at a nodal ring is the pressure times the surface area. Thus, $F_{p_i} = 2\pi\ \Delta z\ r_i\ p_i$. As a numerical example consider a single element solution for a cylinder with an internal pressure of $p = 1$ ksi on the inner radius $r_1 = 10$ in. Assume $E = 10^4$ ksi and $\nu = 0.3$, and let the thickness of the cylinder by 1 in. Note that there is no essential boundary condition on the radial displacement. This is because the hoop effects prevent a rigid body radial motion. The numerical values of the above stiffness contributions are

$$\mathbf{K}_{11} = 2\pi\ \Delta z\ (1.413 \times 10^5) \begin{bmatrix} 1 & -1 \\ -1 & 1 \end{bmatrix}$$

$$\mathbf{K}_{12} = 2\pi\ \Delta z\ (5.769 \times 10^3) \begin{bmatrix} -1 & 0 \\ 0 & 1 \end{bmatrix}, \quad \mathbf{K}_{22} = 2\pi\ \Delta z\ (3.205 \times 10^2) \begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix}$$

while the resultant force at the inner radius is $F_p = 2\pi\ \Delta z\ 10$. Assembling and canceling the common constant gives

$$10^5 \begin{bmatrix} 1.35897 & -1.41026 \\ -1.41026 & 1.47436 \end{bmatrix} \begin{Bmatrix} u_1 \\ u_2 \end{Bmatrix} = \begin{Bmatrix} 10 \\ 0 \end{Bmatrix}.$$

Solving gives $\mathbf{u} = [9.9642 \quad 9.5309] \times 10^{-4}$ in. This represents a displacement error of about 8% and 9%, respectively, at the two nodes. The maximum radial stress equals the applied pressure. The stresses can be found from Eq. 8.14. The hoop strain at node 1 is $\varepsilon_t = u_1 / r_1 = 9.964 \times 10^{-5}$ in/in. The finite element radial strain approximation is $\varepsilon_r = \partial H_1 / \partial r\ u_1 + \partial H_2 / \partial r\ u_2$. The constant radial strain approximation is

$$\varepsilon_r = \frac{-u_1 + u_2}{r_2 - r_1} = -4.333 \times 10^{-4} \text{ in/in}.$$

Therefore, the hoop stress at the first node is

$$\sigma_t = D_{12}\varepsilon_r + D_{22}\varepsilon_t = -2.500 + 13.413 = 10.91 \text{ ksi}.$$

This compares well with the exact value of 10.52 ksi. Note that the inner hoop stress is more than ten times the applied internal pressure. Since we have set ε_z to zero we should

use Eq. 8.11 to determine the axial stress that results from the effect of Poisson's ratio. All three stresses would be used in evaluating a failure criterion like the Von Mises stress.

```
title 'Cylinder with pressure, no spin, Sec. 8.3'      ! 1
axisymmetric ! Problem is axisymmetric, x radius       ! 2
example  110 ! Source library example number           ! 3
data_set  01 ! Data set for example (this file)        ! 4
nodes      3 ! Number of nodes in the mesh             ! 5
elems      1 ! Number of elements in the system        ! 6
dof        1 ! Number of unknowns per node             ! 7
el_nodes   3 ! Maximum number of nodes per element     ! 8
space      1 ! Solution space dimension                ! 9
b_rows     2 ! Number of rows in the B matrix          !10
shape      1 ! Element shape, 1=line, 2=tri, 3=quad    !11
remarks    5 ! Number of user remarks                  !12
gauss      3 ! Maximum number of quadrature point      !13
el_real    3 ! Number of real properties per element   !14
reals      1 ! Number of miscellaneous real properties !15
loads        ! An initial source vector is input       !16
el_no_col    ! All element column matrices are null    !17
no_error_est ! No SCP element error estimates          !18
quit ! keyword input, remarks follow                   !19
1 E = 10e4 ksi, Nu = 0.3, Internal pressure = 1 ksi    !20
2 Resultant = 62.832 kips, Unit axial length           !21
3 U_1_exact = 1.0824e-3   (1)                          !22
4 P = 1 ksi, r=10 in  *----*----*  r=11, P = 0        !23
5            Nodes    1    2    3                     !24
1    0   10.     ! begin nodes, bc flag, r             !25
2    0   10.5                                         !26
3    0   11.                                          !27
 1    1    2    3 ! element                           !28
 1 1.e5 0.3 0.    ! Elem, E, Nu, Rho                    !29
0.0               ! system spin rate                   !30
1    1   62.832   ! node, direction, force             !31
3    1   0.       ! terminate input with last force    !32
```

Figure 8.3.3 Internal pressure load example

```
***  OUTPUT OF RESULTS IN NODAL ORDER  ***              ! 1
    NODE,  Radius r,   DOF_1,                          ! 2
       1  1.0000E+01  9.9667E-04                       ! 3
       2  1.0500E+01  9.7338E-04                       ! 4
       3  1.1000E+01  9.5333E-04                       ! 5
                                                       ! 6
*** STRAIN COMPONENTS AT ELEMENT INTEGRATION POINTS *** ! 7
ELEMENT, PT,  Radius r,  STRAIN_1,   STRAIN_2,         ! 8
      1  1  1.0113E+01 -8.5488E-05  1.0404E-03         ! 9
      1  2  1.0500E+01 -4.8508E-05  9.9792E-04         !10
      1  3  1.0887E+01 -8.3185E-06  9.6295E-04         !11
```

Figure 8.3.4 Displacements and strains due to internal pressure

The implementation of this analysis via numerically integrated elements is shown in Fig. 8.3.2. There we have chosen to include another common loading case of a spinning cylinder (centrifugal load) to account for a body fore vector. The first six lines note that the system is allowing for multiple rows in the arrays **B** and **E**. In solid mechanics the internal volumetric source terms are vectors, unlike the scalar heat source term in the previous example. Even though it has only one component we should always distinguish between scalars and vectors (and higher order tensors), thus line 2 reminds us the system has standard storage space for such optional vectors. The previous hand calculation is repeated here with a single quadratic element. The input data are shown in Fig.8.3.3 which remarks that the exact inner displacement is 1.0824×10^{-3} inches. This gives less than 8 % error in displacements, as noted in the output summary in Fig. 8.3.4, and compares closely to the simple hand estimate. The two strain components are noted to vary significantly over the element and that suggest the mesh is much too crude (as expected). The two average strain values of $[\,-4.744 \;\; 1.000] \times 10^5$ in/in are close to the single linear element results above.

8.4 Exercises

1. In electrostatics the electrical potential, ϕ, is related to the charge density, ζ, and the permittivity of the material, ε. A coaxial cable can be represented in the radial direction by the equation

$$\frac{1}{r}\,\frac{d}{dr}\,(r\,\varepsilon\,\frac{d\phi}{dr}) + \zeta = 0$$

Compare this to Eq. 8.1. A hollow coaxial cable is made from a hollow conducting core and an insulating outer layer with $\varepsilon_1 = 0.5$, $\varepsilon_2 = 2.0$ and charge densities of $\zeta_1 = 100$ and $\zeta_2 = 0$, respectively. The inner, interface, and outer radii are 5, 10, and 25 mm. The corresponding inner and outer potentials (boundary conditions) are $\phi = 500$ and $\phi = 0$, respectively. Compute the interface potential [analytical value is 918.29] using: a) the element formulation in Fig. 8.2.1 for two quadratic elements, b) the approximate closed form integration in Eq.s 8.6 and 7.

2. For the four linear element example in Section 8.2 determine and plot the element centroid fluxes. Obtain an "eyeball" SCP fit of those values to get continuous nodal fluxes and plot them along with the exact flux.

8.5 Bibliography

[1] Buchanan, G.R., *Finite Element Analysis*, New York: McGraw-Hill (1995).

[2] Rockey, K.C., et al., *Finite Element Method − A Basic Introduction*, New York: Halsted Press (1975).

[3] Ross, C.T.F., *Finite Element Programs for Axisymmetric Problems in Engineering*, New York: Halsted Press (1984).

[4] Weaver, W.F., Jr. and Johnston, P.R., *Finite Elements for Structural Analysis*, Englewood Cliffs: Prentice-Hall (1984).
[5] Zienkiewicz, O.C., *The Finite Element Method in Structural and Continuum Mechanics*, New York: McGraw-Hill (1967).

